ÉQUATIONS DIFFÉRENTIELLES p-ADIQUES DANS LES ANNEAUX D'OPÉRATEURS DIFFÉRENTIELS EN INÉGALE CARACTÉRISTIQUE

Alberto Arabia

TABLE DE MATIÈRES

1. MATRICES DE CONNEXION

Soit p un nombre premier, notons $\mathbb{F}_p$ le corps à p éléments et $\mathbb{Z}_p$ l'anneau des entiers p-adiques. Fixons un polynôme $\overline{P} \in \mathbb{F}_p[X_1,\ldots,X_n]$ et notons P un relèvement de $\overline{P}$ dans l'anneau $\mathbb{Z}_p[X_1,\ldots,X_n]$. On notera $\deg(P)$ et $\deg_i(P)$ respectivement : le degré polynôme total de P et celui relatif à la variable X_i.

Soit $\mathbb{Z}_p[\pi]$ l'extension de $\mathbb{Z}_p$ définie par la relation algébrique : $\pi^{p-1}+p=0$ et notons $\mathbf{A}_{p,n}$ l'anneau $\mathbb{Z}_p[\pi][X_1,\ldots,X_n,y^{-1},y]$, où y désigne une variable abstraite. Fixons un entier positif $m > \deg(P)$, appelé *déformation*, et considérons la famille d'opérateurs différentiels $\mathbb{Z}_p[\pi]$-linéaires sur $\mathbf{A}_{p,n}$:

$$\begin{cases} \nabla_i \ : f \to \frac{\partial}{\partial X_i}(f) + \pi \frac{\partial}{\partial X_i}\left(P + y\displaystyle\sum_{k=1}^{n} {X_k}^m\right) f\,, \quad \text{pour } 1 \le i \le n; \\ \nabla_y \ : f \to \frac{\partial}{\partial y}(f) + \pi\left(\displaystyle\sum_{k=1}^{n} {X_k}^m\right) f\,. \end{cases}$$

(*) Université Paris 7-Denis Diderot; 175, rue du Chevaleret, neuvième étage, bureau 9D11, 75013 Paris.
UFR de Mathématiques. Théorie des Groupes, Représentations et Applications.
Adresse électronique : `arabia@mathp7.jussieu.fr`
Téléphone : (1) 44 27 79 57. Fax : (1) 44 27 78 18.

Le module quotient :

$$\mathbf{M}(P,m) := \frac{\mathbf{A}_{p,n}}{\sum_{i=1}^{n} \nabla_i(\mathbf{A}_{p,n})}\ ,$$

est alors un $\mathbb{Z}[\pi, y, y^{-1}]$-module libre de rang $(m-1)^n$ de base canonique : la famille des "monômes" $\prod_{i=1}^{n} {X_i}^{\alpha_i}$, où $0 \leq \alpha_i \leq m-2$. L'opérateur ∇_y opère naturellement sur $\mathbf{M}(P,m)$ et la matrice de la connexion associée, relativement à la base canonique, notée $\mathbf{C}(P,m)$, est une matrice carrée, indépendante de p, à $(m-1)^n$ lignes et colonnes et à coefficients dans l'anneau $\mathbb{Z}_p[\pi, y^{-1}]$.

La matrice $\mathbf{C}(P,m)$ joue un rôle fondamental dans la théorie, mais la réduction des éléments $\nabla_y(\prod_{i=1}^{n} {X_i}^{\alpha_i}) = \pi\Big(\sum_{k=1}^{n} {X_k}^{m}\Big) \prod_{i=1}^{n} {X_i}^{\alpha_i} \in \mathbf{A}_{p,n}$, modulo les relations $\sum_{i=1}^{n} \nabla_i(\mathbf{A}_{p,n})$, est définitivement inaccessible au calcul manuel en dehors de cas triviaux, inintéressants sur le plan théorique. L'informatique est ainsi devenue incontournable pour la détermination explicite des matrices de connexion; la sous-section suivante donne les procédures `Maple` que nous avons rédigées pour leur calcul.

1.1 Procédures `Maple` destinées au calcul de la matrice $\mathbf{C}(P,m)$

```
############################################################################
####################### RELATIONS DE LA CONNEXION #######################
relaciones:=proc() #
local i,k,temp:
global relacion,F,m,n;
 for i from 1 to n do
  relacion[i]:=[]:
  for k from 0 to m-1 do
   relacion[i]:=
               [op(relacion[i]),
                solve(
                collect(
   diff(x[i]^k,x[i])+PPi*(diff(F,x[i])+y*diff(x[i]^m,x[i]))*x[i]^k,
                        x[i]
                        ),
                       x[i]^(k+m-1)
                      )
                 ]:
  od:
 od:
end:
```

> Cette procédure génère la suite
>
> relacion[1],…,relacion[n]
>
> où `relacion[i]` est une liste d'équivalents modulo $\sum_{i=1}^{n} \nabla_i(\mathbf{A}_{p,n})$, des termes ${X_i}^{\alpha_i}$ avec $m-1 \leq \alpha_i \leq 2(m-1)$, pouvant intervenir dans l'explicitation de ∇_y.
> Le degré total, ainsi que les degrés relatifs aux différentes variables X_k, de l'équivalent associé à ${X_i}^{\alpha_i}$, vérifient des majorations garantissant la convergence de la procédure `reduit`.

```
############################################################################
##########################    BASE DE MONOMES     #######################
bon:=proc(Pol) #
local i,flag:
 flag:=true:
 for i from 1 by 1 to n do
  if degree(Pol,x[i]) > m-2 then
   flag:=false:
  fi:
 od:
 RETURN(flag):
end:
```

> Teste la condition $\sup\Big\{\deg_i(\texttt{Pol})\Big\} \leq m-2$

```
############################################################################
##########################  BASE DE M(P,m) ############################
genere_base:=proc() #
local i,h,temp,y:
global Base,card_base,monome,liste_var;
 for i from 1 by 1 to n*(m-2) do
  temp:=monome^(i):
  if n=1 then
   Base:=[op(Base),temp]:
  else
   temp:=sort(expand(temp),liste_var):
   for h from 1 by 1 to nops(temp) do
    y:=op(h,temp):
    if bon(y) then
     Base:=[op(Base),y/tcoeff(y)]:
    fi:
   od:
  fi:
 od:
 card_base:=nops(Base):
 lprint(`Cardinal de la base:`,card_base):
end:
```

> Génère la liste `Base` contenant les éléments $\prod_{i=1}^{n} {X_i}^{\alpha_i}$ de la base canonique de $\mathbf{M}(P,m)$.
> La base est obtenue en developpant
>
> $(X_1+\cdots+X_n)^i$
>
> pour $1 \leq i \leq n(m-2)$ et en receuillant les monômes de bons degrés.

```
##########################################################################
#####################  REDUCTION MODULO RELATIONS  #####################
reduit:=proc(PP) #
local i,temp,dg,L,FLAG:
global relacion,trainnee;
 trainnee:=``:
 temp:=PP:
 FLAG:=1:
 while FLAG = 1 do #
  FLAG:=0:
  L:=[]:
  for i from 1 to n do
   temp:=collect(temp,x[i]):
   L:=[op(L),degree(temp,x[i])]:
  od:
  dg:=max(op(L)):
  if dg > m-2 then
   FLAG:=1:
   member(dg,L,'i'):
   if dg <= 2*(m-1) then
    trainnee:=cat(trainnee,`.`):
    temp:=subs(x[i]^dg=op(dg-(m-2),relacion[i]),temp):
   else
    trainnee:=cat(trainnee,`*`):
    temp:=subs(x[i]^dg=(op(m,relacion[i]))*x[i]^(dg-2*(m-1)),temp):
   fi:
   temp:=collect(expand(temp),liste_var):
  fi:
 od:
 RETURN(temp):
end:
```

Détermine l'élément `reduit(PP)` vérifiant

$$\sup\Big\{\deg_i(\texttt{reduit}(\texttt{PP}))\Big\}\leq m-2\,,$$

équivalent, modulo $\sum_{i=1}^{n}\nabla_i(\mathbf{A}_{p,n})$, d'un élément $\texttt{PP}\in\mathbf{A}_{p,n}$.

Début de la boucle de réduction.
Elle réduit successivement, à l'aide des listes `relacion[i]`, les degrés en chaque variable X_k d'un polynôme `temp`, jusqu'à les rendre tous majorés par $m-2$.

```
##########################################################################
######################  MATRICE DE LA CONNEXION  #######################
######################     PROCEDURE INTERNE     #######################
genere:=proc() #
local i,k,h,temp,b,yy,dg,sec,secc,texte,champ:
global Base,card_base,trainnee,Mat,facteur;
 secc:=time():
 print(`MATRICE DE LA CONNEXION`):
 Mat:=array(1..card_base,1..card_base):
 for i from 1 by 1 to card_base do
  sec:=time():
  temp:=collect(facteur*op(i,Base),list_var):
  temp:=reduit(temp): #
  for h from 1 by 1 to card_base do
   b:=op(h,Base):
   yy:=temp:
   for k from 1 to n do
    dg:=degree(b,x[k]):
    yy:=collect(yy,x[k]):
    yy:=coeff(yy,x[k],dg):
   od:
   Mat[h,i]:=collect(yy,y):
  od:
  sec:=temps(time()-sec):
  champ:=gformat(i,2):
  texte:=cat(`Colonne `,champ,
         ` de la mat. de connexion calculee. Delai:=`,sec,trainnee):
  lprint(texte):
 od:
 secc:=temps(time()-secc):
 lprint(`DELAI TOTAL DE LA  SOUS-ROUTINE:=`,secc):
 Pause():
end:
```

Calcule la matrice de connexion $\texttt{Mat}:=\mathbf{C}(P,m)$.

Appel à la sous-routine de réduction modulo $\sum_{i=1}^{n}\nabla_i(\mathbf{A}_{p,n})$.

```
##########################################################################
######################  MATRICE DE LA CONNEXION  #######################
######################   PROCEDURE PRINCIPALE    #######################
connexion:=proc(a,nn,b) #
local i:
global Base,Mat,monome,facteur,liste_var,F,n,m;
 F:=a:    #polynome
 n:=nn:   #nombre de variables
 m:=b:    #deformation
 Base:=[1]:
 #generation de la liste des variables x[1],...,x[n]
 #et des elements x[1]+...+x[n] et x[1]^m+...+x[n]^m
 liste_var:=[]:
 monome:=0:
 facteur:=0:
 for i from 1 to n do
```

Cette procédure calcule et affiche à l'écran la matrice de connexion $\mathbf{C}(\texttt{a},\texttt{b})$ pour un polynôme $\texttt{a}\in\mathbf{A}_{p,\mathrm{nn}}$ et une déformation $\texttt{b}>\deg(\texttt{a})$.

```
  monome:=monome+x[i]:
  facteur:=facteur+x[i]^m:
  liste_var:=[op(liste_var),x[i]]:
 od:
 facteur:=PPi*facteur:
 # #—————————————————————————————————— Appel des sous-routines principales
 genere_base():
 relaciones():
 interface(quiet=true);
 genere():
 #
 Mat:=scalarmul(Mat,-1):
 Mat:=map(collect,map(collect,Mat,PPi),y):
 if nops(Base)<20 then
  print(Mat): #—————————————————————————————— Affichage la matrice de connexion
 else
  lprint(
  `Trop d'elements dans la base pour afficher la matrice de la connexion !!!`):
 fi:
end:
```

1.2 Exemples numériques de matrices de connexion

Voici quelques explicitations des matrices de connexion $\mathbf{C}(P,m)$ obtenues en exécutant les scripts précédents.

- Pour le polynôme $P = X^3 \in \mathbf{A}_{p,1}$ et les déformations 4, 5, 6, 7, et 8, on obtient :

$$\mathbf{C}(X^3,4) = \begin{bmatrix} \frac{1}{4\,y} & \frac{-3}{16\,y^2} & \frac{9}{64\,y^3} \\ 0 & \frac{1}{2\,y} & \frac{-3}{8\,y^2} \\ \frac{-9\,\pi}{16\,y^2} & \frac{27\,\pi}{64\,y^3} & \frac{-81\,\pi}{256\,y^4} + \frac{3}{4\,y} \end{bmatrix}$$

$$\mathbf{C}(X^3,5) = \begin{bmatrix} \frac{1}{5\,y} & 0 & \frac{-3}{25\,y^2} & 0 \\ 0 & \frac{2}{5\,y} & 0 & \frac{-6}{25\,y^2} \\ 0 & \frac{-9\,\pi}{25\,y^2} & \frac{3}{5\,y} & \frac{27\,\pi}{125\,y^3} \\ \frac{3\,\pi}{5\,y} & 0 & \frac{-9\,\pi}{25\,y^2} & \frac{4}{5\,y} \end{bmatrix}$$

$$\mathbf{C}(X^3,6) = \begin{bmatrix} \frac{1}{6\,y} & 0 & 0 & \frac{-1}{12\,y^2} & 0 \\ 0 & \frac{1}{3\,y} & 0 & 0 & \frac{-1}{6\,y^2} \\ 0 & 0 & \frac{-\pi}{4\,y^2} + \frac{1}{2\,y} & 0 & 0 \\ \frac{\pi}{2\,y} & 0 & 0 & \frac{-\pi}{4\,y^2} + \frac{2}{3\,y} & 0 \\ 0 & \frac{\pi}{2\,y} & 0 & 0 & \frac{-\pi}{4\,y^2} + \frac{5}{6\,y} \end{bmatrix}$$

$$\mathbf{C}(X^3,7) = \begin{bmatrix} \frac{1}{7\,y} & 0 & 0 & 0 & \frac{-3}{49\,y^2} & 0 \\ 0 & \frac{2}{7\,y} & 0 & 0 & 0 & \frac{-6}{49\,y^2} \\ 0 & 0 & \frac{3}{7\,y} & \frac{-9\,\pi}{49\,y^2} & 0 & 0 \\ \frac{3\,\pi}{7\,y} & 0 & 0 & \frac{4}{7\,y} & \frac{-9\,\pi}{49\,y^2} & 0 \\ 0 & \frac{3\,\pi}{7\,y} & 0 & 0 & \frac{5}{7\,y} & \frac{-9\,\pi}{49\,y^2} \\ 0 & 0 & \frac{3\,\pi}{7\,y} & 0 & 0 & \frac{6}{7\,y} \end{bmatrix}$$

$$\mathbf{C}(X^3,8) = \begin{bmatrix} \frac{1}{8\,y} & 0 & 0 & 0 & 0 & \frac{-3}{64\,y^2} & 0 \\ 0 & \frac{1}{4\,y} & 0 & 0 & 0 & 0 & \frac{-3}{32\,y^2} \\ 0 & 0 & \frac{3}{8\,y} & 0 & \frac{-9\,\pi}{64\,y^2} & 0 & 0 \\ \frac{3\,\pi}{8\,y} & 0 & 0 & \frac{1}{2\,y} & 0 & \frac{-9\,\pi}{64\,y^2} & 0 \\ 0 & \frac{3\,\pi}{8\,y} & 0 & 0 & \frac{5}{8\,y} & 0 & \frac{-9\,\pi}{64\,y^2} \\ 0 & 0 & \frac{3\,\pi}{8\,y} & 0 & 0 & \frac{3}{4\,y} & 0 \\ 0 & 0 & 0 & \frac{3\,\pi}{8\,y} & 0 & 0 & \frac{7}{8\,y} \end{bmatrix}$$

- Pour le polynôme $P = X_1 + X_2 + X_3 \in \mathbf{A}_{p,3}$ et les déformations 2 et 3, on obtient :

$\mathbf{C}(X_1 + X_2 + X_3, 2) = \left[\frac{-3\,\pi}{4\,y^2} + \frac{3}{2\,y}\right]$

$$\mathbf{C}(X_1 + X_2 + X_3, 3) = \begin{bmatrix} \frac{1}{y} & \frac{-\pi}{9\,y^2} & \frac{-\pi}{9\,y^2} & \frac{-\pi}{9\,y^2} & 0 & 0 & 0 & 0 \\ \frac{\pi}{3\,y} & \frac{4}{3\,y} & 0 & 0 & \frac{-\pi}{9\,y^2} & \frac{-\pi}{9\,y^2} & 0 & 0 \\ \frac{\pi}{3\,y} & 0 & \frac{4}{3\,y} & 0 & \frac{-\pi}{9\,y^2} & 0 & \frac{-\pi}{9\,y^2} & 0 \\ \frac{\pi}{3\,y} & 0 & 0 & \frac{4}{3\,y} & 0 & \frac{-\pi}{9\,y^2} & \frac{-\pi}{9\,y^2} & 0 \\ 0 & \frac{\pi}{3\,y} & \frac{\pi}{3\,y} & 0 & \frac{5}{3\,y} & 0 & 0 & \frac{-\pi}{9\,y^2} \\ 0 & \frac{\pi}{3\,y} & 0 & \frac{\pi}{3\,y} & 0 & \frac{5}{3\,y} & 0 & \frac{-\pi}{9\,y^2} \\ 0 & 0 & \frac{\pi}{3\,y} & \frac{\pi}{3\,y} & 0 & 0 & \frac{5}{3\,y} & \frac{-\pi}{9\,y^2} \\ 0 & 0 & 0 & 0 & \frac{\pi}{3\,y} & \frac{\pi}{3\,y} & \frac{\pi}{3\,y} & \frac{2}{y} \end{bmatrix}$$

- Pour le polynôme $P = {X_1}^2 + X_2 \in \mathbf{A}_{p,2}$ et déformation 3 et 4 :

$$\mathbf{C}(X_1^2 + X_2, 3) = \begin{bmatrix} \frac{2}{3\,y} & \frac{-2}{9\,y^2} & \frac{-\pi}{9\,y^2} & 0 \\ \frac{-4\,\pi}{9\,y^2} & \frac{8\,\pi}{27\,y^3} + \frac{1}{y} & 0 & \frac{-\pi}{9\,y^2} \\ \frac{\pi}{3\,y} & 0 & \frac{1}{y} & \frac{-2}{9\,y^2} \\ 0 & \frac{\pi}{3\,y} & \frac{-4\,\pi}{9\,y^2} & \frac{8\,\pi}{27\,y^3} + \frac{4}{3\,y} \end{bmatrix}$$

$$\mathbf{C}(X_1^2 + X_2, 4) = \begin{bmatrix} \frac{1}{2\,y} & 0 & 0 & \frac{-1}{8\,y^2} & 0 & \frac{-\pi}{16\,y^2} & 0 & 0 & 0 \\ 0 & \frac{-\pi}{4\,y^2} + \frac{3}{4\,y} & 0 & 0 & 0 & 0 & 0 & \frac{-\pi}{16\,y^2} & 0 \\ \frac{\pi}{4\,y} & 0 & \frac{3}{4\,y} & 0 & 0 & 0 & \frac{-1}{8\,y^2} & 0 & 0 \\ \frac{\pi}{2\,y} & 0 & 0 & \frac{1}{y} - \frac{\pi}{4\,y^2} & 0 & 0 & 0 & 0 & \frac{-\pi}{16\,y^2} \\ 0 & \frac{\pi}{4\,y} & 0 & 0 & \frac{1}{y} - \frac{\pi}{4\,y^2} & 0 & 0 & 0 & 0 \\ 0 & 0 & \frac{\pi}{4\,y} & 0 & 0 & \frac{1}{y} & 0 & 0 & \frac{-1}{8\,y^2} \\ 0 & 0 & \frac{\pi}{2\,y} & \frac{\pi}{4\,y} & 0 & 0 & \frac{-\pi}{4\,y^2} + \frac{5}{4\,y} & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{\pi}{4\,y} & 0 & 0 & \frac{-\pi}{4\,y^2} + \frac{5}{4\,y} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{\pi}{2\,y} & \frac{\pi}{4\,y} & 0 & \frac{-\pi}{4\,y^2} + \frac{3}{2\,y} \end{bmatrix}$$

Remarque : Plusieurs conjectures concernant le comportement asymptotique des matrices $\mathbf{C}(P, m)$, en fonction de m, ont pu être testées à l'aide de ces procédures et de celles de **1.4**.

1.3 Exemple d'affichage lors du calcul des matrices de connexion

Pour $P = {X_1}^2 + X_2$, déformation $m = 4$, le calcul de $\mathbf{C}(X_1^2 + X_2, 4)$ affiche :

```
************************************************************
************************************************************
Cardinal de la base:   9
                            MATRICE DE LA CONNEXION
Colonne  1 de la mat. de connexion calculee. Delai:=0.05s... #
Colonne  2 de la mat. de connexion calculee. Delai:=0.05s....
Colonne  3 de la mat. de connexion calculee. Delai:=0.05s...
Colonne  4 de la mat. de connexion calculee. Delai:=0.05s....
Colonne  5 de la mat. de connexion calculee. Delai:=0.05s....
Colonne  6 de la mat. de connexion calculee. Delai:=0.05s....
Colonne  7 de la mat. de connexion calculee. Delai:=0.067s....
Colonne  8 de la mat. de connexion calculee. Delai:=0.067s.....
Colonne  9 de la mat. de connexion calculee. Delai:=0.066s.....
DELAI TOTAL DE LA  SOUS-ROUTINE:=   0.516s.
```

La suite des points en fin de chaque ligne indique le nombre d'itérations de la boucle de réduction.

1.4 Procédures `Maple` auxiliaires pour l'étude des matrices de connexion

```
##########################################################################
######################    ROUTINES AUXILIAIRES   ######################
diagonale:=proc(deg) #
local i,L:
global card_base,Mat;
 interface(quiet=true):
 L:=[]:
 for i from 1 to card_base do
  L:=[op(L),coeff(Mat[i,i],y,deg)*y^(deg)]:
 od:
 print(L):
 interface(quiet=false):
end:
```

Affiche les termes de degré `deg` en y de la diagonale de la matrice de connexion courante.

```
####################################
sousmat:=proc(dg) #
local i,j,A:
global card_base,Mat;
 interface(quiet=true):
 A:=array(1..card_base,1..card_base):
 for i from 1 to card_base do
  for j from 1 to card_base do
   A[i,j]:=coeff(Mat[i,j],y,dg)*y^dg:
  od:
 od:
 print(A);
 interface(quiet=false):
end:
```

Calcule et affiche la matrice des termes de degré `dg` en y de la matrice de connexion courante.

1.5 Exemples numériques des procédures auxiliaires dans l'étude des matrices de connexion

Pour l'exemple où $P = {X_1}^2 + X_2$ et $m = 3$, on a :

- Procédure : `diagonale`.

$$\texttt{diagonale}(-3) \Rightarrow \left[0, \frac{8\,\pi}{27\,y^3}, 0, \frac{8\,\pi}{27\,y^3}\right].$$

- Procédure : `sousmat`.

$$\texttt{sousmat}(-2) \Rightarrow \begin{bmatrix} 0 & \frac{-2}{9\,y^2} & \frac{-\pi}{9\,y^2} & 0 \\ -\frac{4\,\pi}{9\,y^2} & 0 & 0 & \frac{-\pi}{9\,y^2} \\ 0 & 0 & 0 & -\frac{2}{9\,y^2} \\ 0 & 0 & \frac{-4\,\pi}{9\,y^2} & 0 \end{bmatrix}; \qquad \texttt{sousmat}(-3) \Rightarrow \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & \frac{8\,\pi}{27\,y^3} & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \frac{8\,\pi}{27\,y^3} \end{bmatrix}.$$

2. SOLUTIONS FORMELLES D'ÉQUATIONS DIFFÉRENTIELLES

Pour chaque $t \in \mathbb{Z}_p[\pi]$ différent de 0, on s'intéresse au développement en série entière en $(y - t)$ de la solution de l'équation différentielle

$$\frac{\partial}{\partial y}(F) + \mathbf{C}(P, m) \cdot F = 0\,,$$

où F désigne une matrice carrée à $(m-1)^n$ lignes et colonnes à coefficients dans $\mathbb{Z}_p[\pi][[y-t]]$ vérifiant la condition $F(t) = \mathrm{Id}$. On a donc :

$$F(y) = \sum_{k \geq 0} \frac{1}{k!} \mathbf{C}(P, m, k)(t)(y-t)^k\,,$$

où les matrices $\mathbf{C}(P, m, k)$ vérifient la récurrence :

$$\begin{cases} \mathbf{C}(P, m, 0) = \mathrm{Id}; \\ \mathbf{C}(P, m, 1) = \mathbf{C}(P, m); \\ \mathbf{C}(P, m, k+1) = \frac{\partial}{\partial y}\big(\mathbf{C}(P, m, k)\big) + \mathbf{C}(P, m, k) \cdot \mathbf{C}(P, m). \end{cases}$$

2.1 Procédure `Maple` pour le calcul des matrices $\frac{1}{k!}\mathbf{C}(P,m,k)$

Dans ces procédures, on a noté `G_k` la matrice $\frac{1}{k!}\mathbf{C}(P,m,k)$.

```
############################################################################
############################  MATRICES G_k  ###############################
genere_G:=proc(M,ordre) #
local A,i,B,sec,secc,texte,champ,aux:
global card_base,Liste_G;
 secc:=time():
 A:=matrix(card_base,card_base,0):
 B:=matrix(card_base,card_base,0):
 for i from 1 to card_base do
  A[i,i]:=1:
 od:
 Liste_G:=[eval(A)]:
 aux:=cat(`SUITE DES MATRICES G_n (n=1,...,`,convert(ordre,name),`)`):
 print(aux):
 for i from 1 to ordre do #
  sec:=time():
  B:=map(diff,A,y):
  A:=add(B ,multiply(A,M)):
  A:=map(collect,A,y):
  sec:=temps(time()-sec):
  champ:=dformat_(i,2):
  texte:=cat(`Matrice G_`,champ,` calculee. Delai:=`,sec):
  lprint(texte):
  Liste_G:=[op(Liste_G),eval(scalarmul(A,1/i!))]:
 od: #
 secc:=temps(time()-secc):
 lprint(`DELAI TOTAL DE LA SOUS-ROURINE:=`,secc):
end:
```

Pour `M`=$\mathbf{C}(P,m)$, cette procédure calcule la liste `Liste_G` des matrices : `G_0`,...,`G_ordre`.

Début de la boucle de récursivité pour le calcul des matrices `G_k`.

Fin de la boucle récursivité

2.2 Exemples numériques des matrices $\frac{1}{k!}\mathbf{C}(P,m,k)$

Voici quelques exemples des matrices considérées obtenus à l'aide des procédures précédentes. Dans les sections suivantes, nous traiterons du problème des rayons de convergence de séries définies ci-dessus, le lecteur comprendra aisément la lourdeur des calculs à développer lorsque l'on cherche, par exemple, à calculer les 200 premiers termes des séries en question; ceci a été nécessaire dans les recherches de Mebkhout pour certaines conjectures et a demandé d'importantes ressources matérielles sur Médicis où des scripts nécessitant de plus d'un mois et demi de calculs ininterrompus ont été lancés.

- Pour $P = X^3 \in \mathbf{A}_{p,1}$ et déformation 4, on obtient :

$$\mathbf{C}(X^3,4,0) = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \qquad \mathbf{C}(X^3,4,1) = \begin{bmatrix} \frac{1}{4\,y} & \frac{-3}{16\,y^2} & \frac{9}{64\,y^3} \\ 0 & \frac{1}{2\,y} & -\frac{3}{8\,y^2} \\ -\frac{9\,\pi}{16\,y^2} & \frac{27\,\pi}{64\,y^3} & \frac{-81\,\pi}{256\,y^4} + \frac{3}{4\,y} \end{bmatrix}$$

$$\frac{1}{2}\mathbf{C}(X^3,4,2) = \begin{bmatrix} -\frac{3}{32\,y^2} - \frac{81\,\pi}{2048\,y^5} & \frac{15}{128\,y^3} + \frac{243\,\pi}{8192\,y^6} & \frac{-27}{256\,y^4} - \frac{729\,\pi}{32768\,y^7} \\ \frac{27\,\pi}{256\,y^4} & \frac{-1}{8\,y^2} - \frac{81\,\pi}{1024\,y^5} & \frac{9}{64\,y^3} + \frac{243\,\pi}{4096\,y^6} \\ \frac{9\,\pi}{32\,y^3} + \frac{729\,\pi^2}{8192\,y^6} & \frac{-81\,\pi}{256\,y^4} - \frac{2187\,\pi^2}{32768\,y^7} & \frac{-3}{32\,y^2} + \frac{567\,\pi}{2048\,y^5} + \frac{6561\,\pi^2}{131072\,y^8} \end{bmatrix}$$

$$\frac{1}{3!}\mathbf{C}(X^3,4,3) = \begin{bmatrix} \frac{7}{128\,y^3} + \frac{675\,\pi}{8192\,y^6} + \frac{2187\,\pi^2}{524288\,y^9} & \frac{-47}{512\,y^4} - \frac{2187\,\pi}{32768\,y^7} - \frac{6561\,\pi^2}{2097152\,y^{10}} & \frac{195}{2048\,y^5} + \frac{1701\,\pi}{32768\,y^8} + \frac{19683\,\pi^2}{8388608\,y^{11}} \\ -\frac{81\,\pi}{512\,y^5} - \frac{729\,\pi^2}{65536\,y^8} & \frac{1}{16\,y^3} + \frac{135\,\pi}{1024\,y^6} + \frac{2187\,\pi^2}{262144\,y^9} & \frac{-23}{256\,y^4} - \frac{1701\,\pi}{16384\,y^7} - \frac{6561\,\pi^2}{1048576\,y^{10}} \\ -\frac{123\,\pi}{512\,y^4} - \frac{3645\,\pi^2}{16384\,y^7} - \frac{19683\,\pi^3}{2097152\,y^{10}} & \frac{693\,\pi}{2048\,y^5} + \frac{729\,\pi^2}{4096\,y^8} + \frac{59049\,\pi^3}{8388608\,y^{11}} & \frac{5}{128\,y^3} - \frac{675\,\pi}{2048\,y^6} - \frac{72171\,\pi^2}{524288\,y^9} - \frac{177147\,\pi^3}{33554432\,y^{12}} \end{bmatrix}$$

- Pour $P = X^3 \in \mathbf{A}_{p,1}$ et déformation 5, on obtient :

$$\mathbf{C}(X^3,5,0) = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \qquad \mathbf{C}(X^3,5,1) = \begin{bmatrix} \frac{1}{5\,y} & 0 & \frac{-3}{25\,y^2} & 0 \\ 0 & \frac{2}{5\,y} & 0 & \frac{-6}{25\,y^2} \\ 0 & \frac{-9\,\pi}{25\,y^2} & \frac{3}{5\,y} & \frac{27\,\pi}{125\,y^3} \\ \frac{3\,\pi}{5\,y} & 0 & \frac{-9\,\pi}{25\,y^2} & \frac{4}{5\,y} \end{bmatrix}$$

$$\frac{1}{2}\mathbf{C}(X^3,5,2)=\begin{bmatrix} -\frac{2}{25\,y^2} & \frac{27\,\pi}{1250\,y^4} & \frac{9}{125\,y^3} & \frac{-81\,\pi}{6250\,y^5} \\ -\frac{9\,\pi}{125\,y^3} & \frac{-3}{25\,y^2} & \frac{27\,\pi}{625\,y^4} & \frac{12}{125\,y^3} \\ \frac{81\,\pi^2}{1250\,y^4} & \frac{9\,\pi}{50\,y^3} & \frac{-3}{25\,y^2}-\frac{243\,\pi^2}{6250\,y^5} & -\frac{81\,\pi}{625\,y^4} \\ 0 & \frac{81\,\pi^2}{1250\,y^4} & \frac{9\,\pi}{125\,y^3} & -\frac{2}{25\,y^2}-\frac{243\,\pi^2}{6250\,y^5} \end{bmatrix}$$

$$\frac{1}{3!}\mathbf{C}(X^3,5,3)=\begin{bmatrix} \frac{6}{125\,y^3}-\frac{81\,\pi^2}{31250\,y^6} & \frac{-108\,\pi}{3125\,y^5} & \frac{-34}{625\,y^4}+\frac{243\,\pi^2}{156250\,y^7} & \frac{27\,\pi}{1250\,y^6} \\ \frac{54\,\pi}{625\,y^4} & \frac{8}{125\,y^3}-\frac{81\,\pi^2}{15625\,y^6} & \frac{-36\,\pi}{625\,y^5} & \frac{-38}{625\,y^4}+\frac{243\,\pi^2}{78125\,y^7} \\ -\frac{27\,\pi^2}{250\,y^5} & \frac{-177\,\pi}{1250\,y^4}+\frac{729\,\pi^3}{156250\,y^7} & \frac{7}{125\,y^3}+\frac{2187\,\pi^2}{31250\,y^6} & \frac{72\,\pi}{625\,y^5}-\frac{2187\,\pi^3}{781250\,y^8} \\ -\frac{2\,\pi}{125\,y^3}-\frac{243\,\pi^3}{31250\,y^6} & \frac{-54\,\pi^2}{625\,y^5} & \frac{-6\,\pi}{125\,y^4}+\frac{729\,\pi^3}{156250\,y^7} & \frac{4}{125\,y^3}+\frac{1701\,\pi^2}{31250\,y^6} \end{bmatrix}$$

Remarque : Des conjectures concernant la “complexité” des termes des matrices $\frac{1}{k!}\mathbf{C}(P,m,k)$ en fonction de la déformation, ont pu être testées grâce à de telles explicitations.

2.3 Exemple d’affichage et délais des calculs

A titre d’illustration, l’affichage lors de l’exécution de la procédure `genere_G` pour le calcul des matrices $\frac{1}{k!}\mathbf{C}(X^3,5,k)$ et $k\leq 50$ sur `marie.polytechnique.fr`, a été le suivant :

```
                      SUITE DES MATRICES G_n (n=1,...,50)

Matrice G_1  calculee. Delai:=0.167s.       Matrice G_2  calculee. Delai:=0.066s.
Matrice G_3  calculee. Delai:=0.1s.         Matrice G_4  calculee. Delai:=0.15s.
Matrice G_5  calculee. Delai:=0.184s.       Matrice G_6  calculee. Delai:=0.217s.
Matrice G_7  calculee. Delai:=0.267s.       Matrice G_8  calculee. Delai:=0.483s.
Matrice G_9  calculee. Delai:=0.3s.         Matrice G_10 calculee. Delai:=0.35s.
Matrice G_11 calculee. Delai:=0.45s.        Matrice G_12 calculee. Delai:=0.667s.
Matrice G_13 calculee. Delai:=0.55s.        Matrice G_14 calculee. Delai:=0.816s.
Matrice G_15 calculee. Delai:=0.717s.       Matrice G_16 calculee. Delai:=0.784s.
Matrice G_17 calculee. Delai:=0.9s.         Matrice G_18 calculee. Delai:=1.183s.
Matrice G_19 calculee. Delai:=1.084s.       Matrice G_20 calculee. Delai:=1.5s.
Matrice G_21 calculee. Delai:=1.367s.       Matrice G_22 calculee. Delai:=1.616s.
Matrice G_23 calculee. Delai:=2.267s.       Matrice G_24 calculee. Delai:=2.233s.
Matrice G_25 calculee. Delai:=2.316s.       Matrice G_26 calculee. Delai:=2.567s.
Matrice G_27 calculee. Delai:=2.867s.       Matrice G_28 calculee. Delai:=3.15s.
Matrice G_29 calculee. Delai:=3.616s.       Matrice G_30 calculee. Delai:=4.483s.
Matrice G_31 calculee. Delai:=5.684s.       Matrice G_32 calculee. Delai:=7.35s.
Matrice G_33 calculee. Delai:=5.767s.       Matrice G_34 calculee. Delai:=8.184s.
Matrice G_35 calculee. Delai:=8.283s.       Matrice G_36 calculee. Delai:=9.3s.
Matrice G_37 calculee. Delai:=8.05s.        Matrice G_38 calculee. Delai:=8.45s.
Matrice G_39 calculee. Delai:=8.75s.        Matrice G_40 calculee. Delai:=8.566s.
Matrice G_41 calculee. Delai:=9.167s.       Matrice G_42 calculee. Delai:=9.884s.
Matrice G_43 calculee. Delai:=10.55s.       Matrice G_44 calculee. Delai:=12.117s.
Matrice G_45 calculee. Delai:=12.4s.        Matrice G_46 calculee. Delai:=13.083s.
Matrice G_47 calculee. Delai:=14.85s.       Matrice G_48 calculee. Delai:=15.267s.
Matrice G_49 calculee. Delai:=15.866s.      Matrice G_50 calculee. Delai:=17.084s.

DELAI TOTAL DE LA SOUS-ROURINE:=   4m. 34.8s.
```

3. NORMES ET RAYON DE CONVERGENCE

On considère sur $\mathbb{Z}_p$ la valuation p-adique. La relation algébrique $\pi^{p-1}+p=0$ induit alors une structure d’anneau normé sur $\mathbb{Z}_p[\pi]$ par :

$$\left|\sum_{k=0}^{p-2} a_k\,\pi^k\right|_p := \sup\left\{|a_k|_p\cdot p^{\frac{-k}{p-1}}\right\}\quad \text{où } a_k\in\mathbb{Z}_p\,.$$

On définit alors, pour chaque nombre réel strictement positif r appelé *rayon*, une norme $\|\cdot\|_r$ sur le corps $\mathbb{Z}_p[\pi](t)$ par le procédé suivant. Pour tout $Q(t) \in \mathbb{Z}_p[\pi](t)$, sous la forme $Q(t) = \frac{A(t)}{B(t)}$, où $A(t)$ et $B(t)$ appartiennent à $\mathbb{Z}_p[\pi, t]$, on pose :

$$\|Q(t)\|_r := \frac{\|A(t)\|_r}{\|B(t)\|_r} \,,$$

puis, si $A(t) = \sum a_k\, t^k$, avec $a_k \in \mathbb{Z}_p[\pi]$, on pose :

$$\|A(t)\|_r := \sup\left\{|a_k|_p \cdot r^k\right\} \,,$$

en particulier $\|t\|_r = r$.

Les matrices $\frac{1}{k!}\mathbf{C}(P,m,k)(t)$ ont leurs coefficients dans le corps $\mathbb{Z}_p[\pi](t)$ de sorte que les séries formelles de la section précédente auront des rayons de convergence dépendant du rayon r choisi. Un partie significative des recherches de Mebkhout concerne précisément le comportement asymptotique de ces rayons de convergence lorsque $r \to 1^-$.

D'un point de vue théorique, il était important de pouvoir calculer les rayons de convergence suivant des directions $\vec{V} \in \mathbb{Z}_p{}^{(m-1)^n}$ bien choisies et, en particulier, suivant les directions des vecteurs $\vec{e_i}$ de la base canonique.

Notons $\mathbf{R}(P,m,p,r,\vec{V})$ le rayon de convergence de la série formelle :

$$\sum_{k\geq 0} \frac{1}{k!}\mathbf{C}(P,m,k)(t)(y-t)^k \cdot \vec{V} \,.$$

On a la formule de Cauchy bien connue :

$$\mathbf{R}(P,m,p,r,\vec{V}) = \limsup_{k\to\infty} \mathbf{R}_k(P,m,p,r,\vec{V}) \,, \quad \text{où} \quad \mathbf{R}_k(P,m,p,r,\vec{V}) := \sqrt[k]{\frac{\|k!\|_p}{\|\mathbf{C}(P,m,k)(t)\cdot\vec{V}\|_r}} \,.$$

Un résultat intéressant à souligner relatif au comportement des nombres $\mathbf{R}_k(P,m,p,r,\vec{V})$ affirme que lorsque $r = 1$, $\vec{V} \neq 0$ et que p est relativement premier à m, on a

$$\mathbf{R}_k(P,m,p,r,\vec{V}) \leq 1 \,, \quad \text{pour tout } k >> 0 \,,$$

et $\lim_{k\to\infty} \mathbf{R}_k(P,m,p,r,\vec{V}) = 1$.

L'étude des nombres réels $\mathbf{R}_k(P,m,p,r,\vec{V})$ en fonction des six paramètres et suivant diverses évolutions asymptotiques, a constitué une bonne partie des recherches numériques de la première moitié de l'année 1994.

3.1 Procédures `Maple` pour le calcul des normes

Les variables globales dans ce script sont `NNT`, `NNPPi` et `infty`, la première désigne $\|t\|_r = r$, puis `NNPPi` $= \|\pi\|_r = |\pi|_p = p^{\frac{-1}{p-1}}$, où p est la caractéristique en cours, enfin `infty` est la valuation p-adique de 0 que nous avons fixé à 1000 pour les besoins des calculs.

```
##############################################################################
#######################      NORME p-adique        ######################
NN:=proc(Q) #
 RETURN(Poly_NN(numer(Q))/Poly_NN(denom(Q))):
end:
######################################
Poly_NN:=proc(PP) #
local ld,md,i,L,P:
global NNT;
 P:=collect(PP,y):
 ld:=ldegree(P,y):
 md:=degree(P,y):
 L:=[]:
 for i from ld by 1 while i <= md do
  L:=[op(L),evalf(PPi_NN(coeff(P,y,i))*NNT^i)]:
```

Détermine les numérateur et dénominateur de la fraction rationnelle `Q`$\in\mathbb{Z}_p(y,\pi)$, en tant qu'éléments de $\mathbb{Z}_p[y,\pi]$, et rend le quotient des normes p-adiques respectives.

Pour tout polynôme `PP`$\in\mathbb{Z}_p[y,\pi]$, écrit sous la forme :

$$\mathtt{PP}=\sum_{i\geq 0} a_i\, y^i \,, \quad \text{où } a_i\in\mathbb{Z}_p[\pi],$$

cette procédure rend le nombre réel :

$$\mathtt{Poly\_NN(PP)}=\sup\left\{|a_i|_p\cdot|y|_p^i\right\}_{i\geq 0} .$$

```
 od:
 RETURN(max(op(L))):
end:
####################################
PPi_NN:=proc(P) #
```

Pour tout polynôme $\mathtt{P}\in\mathbb{Z}_p[\pi]$, écrit sous la forme :

$$\mathtt{P}=\sum_{i\geq 0} m_i\,\pi^i\,,\quad \text{où } m_i\in\mathbb{Z},$$

cette procédure rend le nombre réel :

$$\mathtt{PPi\_NN}(\mathtt{P})=\sup\Big\{|m_i|_p\cdot|\pi|_p^i\Big\}_{i\geq 0}.$$

```
local ld,md,L,i,P_mod;
global NNPPi;
 P_mod:=collect(rem(P,PPi^(p-1)+p,PPi),PPi):
 ld:=ldegree(P_mod,PPi):
 md:=degree(P_mod,PPi):
 L:=[]:
 for i from ld by 1 while i <= md do
  L:=[op(L),evalf(Z_NN(coeff(P_mod,PPi,i))*NNPPi^i)]:
 od:
 RETURN(max(op(L))):
end:
####################################
Z_NN:=proc(n) #
```

Pour $n\in\mathbb{Z}$, rend la norme p-adique de n

```
local val;
global infty:
 if (n=0)
  then RETURN(0);
 else val:=gcd(n,p^infty);
  if (val=p^infty) then print(`norme_infinie_rencontree`) fi;
  RETURN(1/val);
 fi;
end:
```

3.2 Procédure `Maple` pour le calcul des rayons de convergence

```
#########################################################################
#######################    RAYONS DE CONVERGENCE  #######################
rayon:=proc(S,indice) #
```

Pour tout vecteur $\mathtt{S}\in\mathbb{Z}_p[\pi](t)^{(m-1)^n}$, cette procédure calcule le nombre réel $\|S\|_r^{\frac{-1}{\mathtt{indice}}}$. Un message est affiché lorsque ce nombre est ∞.

```
local j,n,sec,NNS,LS,ray,texte,champ_1,champ_2;
 sec:=time():
 LS:=convert(S,list):
 NNS:=map(NN,LS):
 n:=max(op(NNS)):
 sec:=temps(time()-sec):
 if n=0 then
  lprint(`LA NORME DE F`,indice,`EST NULLE !!!`,`Delai:=`,sec):
 else
  ray:=convert(evalf(r),name):
  champ_1:=dformat_(indice,2):
  champ_2:=fformat(evalf(n^(1/indice))^(-1),12):
  texte:=cat(`Rayon de F_`,champ_1,`:=`,champ_2,` r:=`,ray,` Delai:=`,sec):
  lprint(texte):
 fi:
end:
#########################################################################
#######################     ROUTINE PRINCIPALE    #######################
work:=proc(Poly,var,def,rr,pp,ordre) #
```

Procédure principale destinée au calcul des rayons de convergence des séries formelles développées à l'ordre `ordre` et associées à la matrice de connexion $\mathbf{C}$(`Poly`,`def`), où `Poly`$\in\mathbf{A}_{\mathrm{pp,var}}$ et où le rayon vaut r=`rr`.

```
local i,j,flag,rep,sec,total_time,texte,aux;
global r,p,NNT,NNPPi,last_command,card_base,Mat,Liste_G;
 total_time:=time():
 etoiles(60,`*`):
 etoiles(60,`*`):
 interface(quiet=true);
 r:=convert(rr,rational):
 p:=pp:
 NNT:=r;
 NNPPi:=p^(-1/(p-1));
 connexion(Poly,var,def): #
```

Calcul de la matrice de connexion $\mathbf{C}$(`Poly`,`def`).

```
 Pause():
 genere_G(Mat,ordre): #
```

Génération des matrices $\frac{1}{k!}\mathbf{C}$(`Poly`,`def`,k) pour $k\leq$`ordre`.

```
 interface(quiet=false):
 lprint(``):
 rep:=readstat(
 `CALCUL DES RAYONS SUIVANT LES DIRECTIONS CANONIQUES? [1,0]: `):
 if rep=1 then
  interface(quiet=true):
  lprint(` `):
  lprint(`SUITE DES RACINES N-IEMES DES INVERSES DES NORMES DES F_N`):
  for j from 1 to card_base do #
```

Début de la boucle donnant successivement, pour chaque vecteur $\vec{e}_{\mathrm{j}}$ de la base canonique, la suite de nombres $\mathbf{R}_k$(`Poly`,`def`,`pp`,`rr`,$\vec{e}_{\mathrm{j}}$), pour $k\leq$`ordre`.

```
   lprint(` `):
   texte:=cat(`VECTEURS F_n POUR LA CONDITION INITIALE E_`,convert(j,name)):
   lprint(texte):
   aux:=cat(`E_`,convert(j,name),`:=[`):
    for i from 1 to card_base-1 do
```

```
      if i=j then
       aux:=cat(aux,`1,`):
      else
       aux:=cat(aux,`0,`):
      fi:
     od:
      if i=j then
       aux:=cat(aux,`1]`):
      else
       aux:=cat(aux,`0]`):
     fi:
    lprint(aux):
    sec:=time():
     for i from 2 to ordre+1 do
      rayon(col(op(i,Liste_G),j),i-1): #
     od:
    sec:=temps(time()-sec):
    lprint(`DELAI TOTAL:=`,sec):
   od:
  fi:
  total_time:=temps(time()-total_time):
  total_time:=cat(`| TEMPS TOTAL DE CALCUL POR "work":= `,total_time,` |`):
  etoiles_(length(total_time)-1,`-`):
  lprint(total_time):
  etoiles_(length(total_time)-1,`-`):
  Prompt():
end:
```

Calcule et affiche les nombres $\mathbf{R}_k$(Poly,def,pp,rr,$\vec{e}_j$) pour $1 \leq k \leq$ordre.

3.3 Exemple numérique de calcul des rayon de convergence

- Pour le polynôme $P = X^3 \in \mathbf{A}_{3,1}$, déformation $m = 4$, rayon $r = 0.95$ et jusqu'à l'ordre 20, nous avons obtenu les résultats suivants sur `marie.polytechnique.fr` :

```
SUITE DES RACINES N-IEMES DES INVERSES DES NORMES DES F_N #
#
VECTEURS F_n POUR LA CONDITION INITIALE E_1
E_1:=[1,0,0] #
Rayon en F_1 := 0.949999999999 r:=0.95 Delai:=0.134s.
Rayon en F_2 := 1.645448267190 r:=0.95 Delai:=0.184s.
Rayon en F_3 := 0.949999999999 r:=0.95 Delai:=0.3s.
Rayon en F_4 := 0.949999999999 r:=0.95 Delai:=0.367s.
Rayon en F_5 := 1.183444392634 r:=0.95 Delai:=0.483s.
Rayon en F_6 := 0.949999999999 r:=0.95 Delai:=0.55s.
Rayon en F_7 := 0.949999999999 r:=0.95 Delai:=0.7s.
Rayon en F_8 := 1.152430495111 r:=0.95 Delai:=0.783s.
Rayon en F_9 := 0.992669953649 r:=0.95 Delai:=0.867s.
Rayon en F_10:= 0.988318101205 r:=0.95 Delai:=1.3s.
Rayon en F_11:= 0.989374428781 r:=0.95 Delai:=1.167s.
Rayon en F_12:= 0.981826068512 r:=0.95 Delai:=1.533s.
Rayon en F_13:= 0.979340503288 r:=0.95 Delai:=1.367s.
Rayon en F_14:= 0.980801921482 r:=0.95 Delai:=1.8s.
Rayon en F_15:= 1.221901434890 r:=0.95 Delai:=1.984s.
Rayon en F_16:= 1.175217089190 r:=0.95 Delai:=1.716s.
Rayon en F_17:= 1.648872735270 r:=0.95 Delai:=2.267s.
Rayon en F_18:= 0.992669953649 r:=0.95 Delai:=2.3s.
Rayon en F_19:= 0.990377121657 r:=0.95 Delai:=2.517s.
Rayon en F_20:= 0.990856058885 r:=0.95 Delai:=2.65s.
DELAI TOTAL:=   26.017s.
#
VECTEURS F_n POUR LA CONDITION INITIALE E_2
E_2:=[0,1,0] #
Rayon en F_1 := 0.949999999999 r:=0.95 Delai:=0.116s.
Rayon en F_2 := 0.949999999999 r:=0.95 Delai:=0.2s.
Rayon en F_3 := 0.933895193866 r:=0.95 Delai:=0.3s.
Rayon en F_4 := 0.937895617656 r:=0.95 Delai:=0.383s.
Rayon en F_5 := 0.949999999999 r:=0.95 Delai:=0.483s.
Rayon en F_6 := 0.941913177619 r:=0.95 Delai:=0.833s.
Rayon en F_7 := 0.943064209619 r:=0.95 Delai:=0.633s.
Rayon en F_8 := 1.898425559170 r:=0.95 Delai:=0.75s.
Rayon en F_9 := 0.944601107261 r:=0.95 Delai:=1.1s.
Rayon en F_10:= 0.945139612952 r:=0.95 Delai:=0.966s.
Rayon en F_11:= 0.949999999999 r:=0.95 Delai:=1.233s.
Rayon en F_12:= 0.945947947161 r:=0.95 Delai:=1.167s.
Rayon en F_13:= 0.946259028983 r:=0.95 Delai:=1.367s.
Rayon en F_14:= 0.949999999999 r:=0.95 Delai:=1.467s.
Rayon en F_15:= 1.187006797620 r:=0.95 Delai:=2.033s.
Rayon en F_16:= 1.142649295150 r:=0.95 Delai:=2.s.
Rayon en F_17:= 1.372690295210 r:=0.95 Delai:=1.967s.
```

On note F_N le vecteur $\frac{1}{\mathtt{N}!}\mathbf{C}(P,m,\mathtt{N})(t)\cdot\vec{V}$

Cas où la direction $\vec{V}$ est donnée par le vecteur $\vec{e}_1$=[1,0,0] de la base canonique

Cas où la direction $\vec{V}$ est donnée par le vecteur $\vec{e}_2$=[0,1,0] de la base canonique

```
Rayon en F_18:= 0.968337305571 r:=0.95 Delai:=2.484s.
Rayon en F_19:= 0.967363419055 r:=0.95 Delai:=2.533s.
Rayon en F_20:= 0.968969656978 r:=0.95 Delai:=2.666s.
DELAI TOTAL:=   25.35s.
```

3.4 Procédures complémentaires dans l'étude des séries formelles

Une fois la liste des matrices $\frac{1}{k!}\mathbf{C}(P,m,k)$ calculée, trois routines `Maple` ont été utilisées pour des études complémentaires; les voici :

```
#########################################################################
###########################       DEGRES      ###########################
degres:=proc() #
```

Donne la liste des y-valuations des matrices $\frac{1}{k!}\mathbf{C}(P,m,k)$.

```
local i,L,AUX:
global Liste_G;
 interface(quiet=true):
 L:=[]:
 for i from 1 to nops(Liste_G) do
  AUX:=map(ldegree,op(i,Liste_G),y):
  AUX:=convert(AUX,set):
  L:=[op(L),min(op(AUX))]:
 od:
 print(L):
 interface(quiet=false):
end:
#########################################################################
###########################      SOLUTION     ###########################
solution:=proc(i) #
```

Calcule la somme de la série $\sum_{k=0}^{\texttt{ordre}} \frac{z^k}{k!}\mathbf{C}(P,m,k)(y)\cdot\vec{e}_{\texttt{i}}$

```
local vec,j;
global card_base,Liste_G;
 interface(quiet=true);
 vec:=matrix(card_base,1,0):
 for j from 1 to nops(Liste_G) do
  vec:=add(vec,scalarmul(col(op(j,Liste_G),i),z^(j-1))):
 od:
 interface(quiet=false);
 RETURN(vec):
end:
#########################################################################
###########################     DIRECTION     ###########################
direction:=proc(vecteur) #
```

Calcule et affiche la suite $\mathbf{R}_k(P,m,p,r,\vec{V})$ pour une direction $\vec{V}$ donnée.

```
local i,j,sec,flag:
global r,NNT,NNPPi,card_base,Liste_G,INI;
 sec:=time():
 r:=convert(r,rational):
 NNT:=r;
 NNPPi:=p^(-1/(p-1));
 interface(quiet=true);
 INI:=matrix(card_base,1,0):
 for i from 1 by 1 to card_base do
  flag:=0:
  for j from 1 by 1 to nops(vecteur)/2 do
   if i=op(2*j-1,vecteur) then
    flag:=convert(op(2*j,vecteur),rational):
   fi:
  od:
  INI[i,1]:=flag:
 od:
 lprint(` `):
 lprint(`Caracteristique:=`,p):
 lprint(`Rayon:=`,r):
 lprint(`Condition Initiale`):
 print(convert(convert(col(INI,1),list),name)):
 print(`SUITE DES RACINES N-IEMES DES INVERSES DES NORMES DES F_N`):
 for i from 2 to nops(Liste_G) do
  rayon(col(multiply(op(i,Liste_G),INI),1),i-1):
 od:
 sec:=temps(time()-sec):
 lprint(`DELAI TOTAL DE LA SOUS-ROUTINE:=`,sec):
 interface(quiet=false);
end:
```

3.5 Exemples numériques d'études complémentaires des séries formelles

Appliquées au cas $P = X^3 \in \mathbf{A}_{3,1}$, $m = 4$, $r = 0.95$, $p = 3$ et jusqu'à l'ordre 20, les procédures précédentes donnent :

- Procédure `degres`.

```
> degres();

 [0, -4, -8, -12, -16, -20, -24, -28, -32, -36, -40, -44, -48, -52, -56, -60,
     -64, -68, -72, -76, -80]
```

- Procédure `solution`. Cette routine, utile dans certaines recherches théoriques, ne prévoit pas d'affichage à l'écran. En effet, dans l'exemple en cours, l'expression du vecteur solution `solution(1)` comporte respectivement 210, 211 et 190 termes dans ses trois coordonnées. La fin de l'expression de la troisième coordonnée étant alors :

```
                                                                      18
               717099303999651277050072662753116647                PPi
       + ---------------------------------------------------------- -----
         4134699571508970836945737422722859279186881928935505920000   72
                                                                     y
                                                                       19
                 225593026993218070242020685165761 7                 PPi
       + ----------------------------------------------------------- -----
         13231038628828706678226359752713149693398022172593618944000   75
                                                                      y
                                                                          20
                  250315550499324160131557198608584 9                   PPi
       + -------------------------------------------------------------- -----)]
         33871458889801489096259480966945663215098936761839664496640000   78
                                                                         y
```

4. VECTEURS CYCLIQUES, POLYNÔMES INDICIELS ET EXPOSANTS FORMELS

Dans l'étude des solutions formelles les objets considérés dans cette section jouent un rôle fondamental. Rappelons que dans la section **1** nous avons introduit le $\mathbb{Z}[\pi, y, y^{-1}]$-module libre $\mathbf{M}(P, m)$ de rang $N := (m-1)^n$. Tout élément $\xi \in \mathbf{M}(P, m)$, tel que la famille :

$$\xi,\ (y\nabla_y)(\xi),\ (y\nabla_y)^2(\xi),\ \ldots,\ (y\nabla_y)^{N-1}(\xi)\,, \tag{†}$$

est $\mathbb{Z}[\pi, y, y^{-1}]$-linéairement indépendante sera appelé *vecteur cyclique pour la connexion* $y\nabla_y$. Notons $\mathbb{Z}_p(\pi, y)$ le corps des fractions de $\mathbb{Z}[\pi, y, y^{-1}]$, le vecteur ξ est donc cyclique, *si et seulement si*, la famille (†) est une base du $\mathbb{Z}_p(\pi, y)$-espace vectoriel $\mathbb{Z}_p(\pi, y) \otimes \mathbf{M}(P, m)$. Des résultats théoriques sur les équations différentielles p-adiques garantissent l'existence de tels vecteurs, mais ceci de manière non effective. Nous indiquerons dans la section **4.1** une procédure `Maple` destinée à les détecter.

Fixons maintenant un élément $\xi_0 \in \mathbf{M}(P, m)$ cyclique pour la connexion $y\nabla_y$ et considérons la décomposition :

$$(y\nabla_y)^N(\xi) = A_0\,\xi + A_1\,(y\nabla_y)(\xi) + A_2\,(y\nabla_y)^2(\xi) + \cdots + A_{N-1}\,(y\nabla_y)^{N-1}(\xi)\,,$$

où $A_k \in \mathbb{Z}_p[\pi](y)$. Remarquons que les coefficients A_k sont précisément ceux de la dernière colonne de la matrice de la connexion $y\nabla_y$ relative à la base des $(y\nabla_y)^i(\xi)$ de l'espace vectoriel $\mathbb{Z}_p(\pi, y) \otimes \mathbf{M}(P, m)$. Cette matrice est donc la matrice *compagnon* :

$$((y\nabla_y)) = \begin{pmatrix} 0 & \cdots & \cdots & \cdots & A_0 \\ 1 & 0 & \cdots & \cdots & A_1 \\ 0 & 1 & 0 & \cdots & A_2 \\ \vdots & \vdots & \ddots & \ddots & \vdots \\ 0 & \cdots & \cdots & 1 & A_{N-1} \end{pmatrix}$$

Il existe alors des éléments $B_0, \ldots, B_N \in \mathbb{Z}_p[\pi][y]$, tels que l'on ait :

$$B_N\,(y\nabla_y)^N(\xi) = B_0\,\xi + B_1\,(y\nabla_y)(\xi) + B_2\,(y\nabla_y)^2(\xi) + \cdots + B_{N-1}\,(y\nabla_y)^{N-1}(\xi)\,,$$

Notons $d_{\min}$ et $d_{\max}$ respectivement les bornes inférieure et supérieure des degrés en y intervenant dans les différents B_k. Soient $(B_k)_{(0)}$ et $(B_k)_{(\infty)}$ les coefficients des termes de B_k dont les degrés en y valent respectivement $d_{\min}$ et $d_{\max}$ et posons :

$$\left\{\begin{array}{lcl} \mathrm{In}_{\xi_0}(P,m,0) & = & B_{N\,(0)}\Theta^N - \displaystyle\sum_{k=0}^{N-1} B_{k\,(0)}\Theta^k\,, \\ \mathrm{In}_{\xi_0}(P,m,\infty) & = & B_{N\,(\infty)}\Theta^N - \displaystyle\sum_{k=0}^{N-1} B_{k\,(\infty)}\Theta^k\,. \end{array}\right.$$

Ce sont les *polynômes indiciels*, respectivement *à l'origine et à l'infini*, associés au couple (P,m). Leurs racines (en Θ), dans la cloture algébrique de $\mathbb{Z}_p[\pi]$, sont appelées les *exposants formels* de la connexion $y\nabla_y$.

Rappelons que si les polynômes In_ξ dépendent clairement du choix de l'élément cyclique ξ, un théorème affirme que les classes modulo $\mathbb{Z}$ des exposants en sont par contre indépendantes.

4.1 Procédure `Maple` pour la détection d'un vecteur cyclique

L'idée implémentée par les procédures suivantes est élémentaire, elle répose sur le fait que $\xi \in \mathbf{M}(P,m)$ est cyclique, si et seulement si, le déterminant de la matrice à N lignes et N colonnes dont les coefficients représentent les coefficients de $(y\nabla_y)^k(\xi)$ relativement à la base canonique de $\mathbf{M}(P,m)$, pour $k = 0, \ldots, N-1$, est non nul. On calcule alors ces déterminant sur des polynômes de $\mathbb{Z}_p[X_1, \ldots, X_N, y]$, de plus en plus compliqués, jusqu'à obtention d'un résultat non nul. Notons $\{\vec{e}_0 = 1, \vec{e}_1, \ldots, \vec{e}_{N-1}\}$ la base canonique de $\mathbf{M}(P,m)$, les polynômes en question sont alors obtenus par la formule :

$$\xi(s,t) = \sum_{j=0}^{N-1}\left(\sum_{i=0}^{s-1} a(i,j)\,y^i\right)\vec{e}_j + \sum_{j=0}^{t} a(s,j)\,y^s\,\vec{e}_j\,,$$

en faisant varier t entre 0 et $N-1$, pour chaque $s = 0, 1, 2, \ldots$ (Dans ces expressions les termes $a(i,j)$ désignent des constantes abstraites.)

Bien entendu, si ces idées sont d'une grande simplicité théorique, les calculs nécessaires à leur explicitation sont décidemment inaccessibles à l'être humain et le choix informatique reste, une fois de plus, la seule alternative envisageable.

```
############################################################################
###########################    COORDONNEES    ###########################
coordonnees:=proc(Poly) #
local temp,tempp,L,b,h,k,dg:
global card_base,Base:
 L:=[]:
 temp:=reduit(Poly):
  for h from 1  to card_base do
   b:=op(h,Base):
   tempp:=temp:
   for k from 1 to nops(liste_var) do
    dg:=degree(b,x[k]):
    tempp:=collect(expand(tempp),x[k]):
    tempp:=coeff(tempp,x[k],dg):
   od:
   L:=[op(L),tempp]:
  od:
 RETURN(L):
end:
```

Calcule pour un élément $\texttt{Poly}{\in}\mathbf{A}_{p,n}$ la liste des coordonnées canoniques de l'élément de $\mathbf{M}(P,m)$ qu'il représente.

```
##########################################################################
###########################    DET_CYCLIQUE   ############################
det_cyclique:=proc(Poly) # Génère la matrice des coordonnées canoniques des éléments (y∇_y)^k(Poly), pour k=0,...,N-1; puis calcule son déterminant.
local text,temp,tempp,j,L_Poly,dg,sec:
global card_base,facteur,Base,liste_var;
 temp:=reduit(Poly):
 L_Poly:=[temp]:
 for j from 1 to card_base-1 do
  sec:=time():
  temp:=reduit(y*diff(temp,y)+y*facteur*temp):
  L_Poly:=[op(L_Poly),temp]:
  sec:=temps(time()-sec):
  text:=cat(`Itere `,j+1,`/`,card_base,` calcule. Delai:=`,sec):
  lprint(text):
 od:
 sec:=time():
 L_Poly:=map(coordonnees,L_Poly):
 temp:=simplify(det(L_Poly)):
 sec:=temps(time()-sec):
 lprint(`Delai de calcul du determinant:=`,sec):
 RETURN(temp):
end:
##########################################################################
######################## RECHERCHE-CYCLIQUE  ###########################
trouve_fonction_cyclique:=proc() # Calcule les valeurs det_cyclique(ξ(s,t)) jusqu'à obtention d'un résultat non nul. Affecte à la variable globale "cyclique" le premier élément ξ(s,t) satisfaisant à cette dernière condition.
 local i,h,k,Poly,determinant,total_time,sols:
 global card_base,Base,liste_var,Liste_A,cyclique;
 interface(quiet=true);
  total_time:=time():
  Poly:=0:
  determinant:=0:
  Liste_A:=[]:
  for h from 0 while determinant = 0 do # Boucle génératrice des polynômes Poly :=ξ(h,k)
   for k from 1 while (determinant = 0 and k <= card_base) do:
   Liste_A:=[op(Liste_A),a[k-1,h]]:
   Poly:=Poly+a[k-1,h]*y^(h)*op(k,Base):
   Poly:=sort(Poly,liste_var):
   lprint(`----------`):
   lprint(`Teste la famille de polynomes`):
   print(Poly):
   determinant:=det_cyclique(Poly): # Teste si Poly est cyclique
   if determinant = 0 then
    lprint(`ECHEC !!`):
   fi:
od:
 od: # Fin de la boucle génératice des ξ(h,k)
 etoiles(60,`*`):
 lprint(`La famille suivante comporte des elements cycliques`):
 lprint(` pour la connexion "y Nabla_y".`):
 lprint():
 print(Poly):
 lprint(`Elle est enregistree dans la variable globale de nom "cyclique"`):
 total_time:=temps(time()-total_time):
 lprint(`Delai de la recherche:=`,total_time):
 etoiles(60,`*`):
 cyclique:=Poly:
 print(`CONDITIONS DE NON CYCLICITE`): # Recheche des conditions sur les coefficients a(k,h) donnant des polynômes ξ(h,k) non cycliques.
 sols:=solve({0=determinant},{op(Liste_A)}):
 print(sols):
 etoiles(60,`*`):
 interface(quiet=false):
end:
##########################################################################
macro(vect_cyclique=trouve_fonction_cyclique()): # Raccourci d'appel de procédure
##########################################################################
```

4.2 Exemples numériques de vecteurs cycliques

- Pour le polynôme $P = X^3 \in \mathbf{A}_{p,1}$ et la déformation 8, la procédure `vect_cyclique` donne l'affichage :

```
----------
Teste la famille de polynomes
                                   a[0, 0]
Itere 2/7 calcule. Delai:=0.033s.
Itere 3/7 calcule. Delai:=0.017s.
Itere 4/7 calcule. Delai:=0.083s.
Itere 5/7 calcule. Delai:=0.1s.
```

```
Itere 6/7 calcule. Delai:=0.134s.
Itere 7/7 calcule. Delai:=0.167s.
Delai de calcul du determinant:=   0.65s.
************************************************************
La famille suivante comporte des elements cycliques
 pour la connexion "y Nabla_y".
                                    a[0, 0]
Elle est enregistree dans la variable globale de nom "cyclique"
Delai de la recherche:=   1.217s.
************************************************************
                          CONDITIONS DE NON CYCLICITE
  {a[0, 0] = 0}, {a[0, 0] = 0}, {a[0, 0] = 0}, {a[0, 0] = 0}, {a[0, 0] = 0},
      {a[0, 0] = 0}, {a[0, 0] = 0}
************************************************************
```

L'élément $\xi = 1$ est donc bien cyclique dans $\mathbf{M}(X^3, 8)$.

On aura remarqué que la condition $a(0,0) = 0$ apparaît avec multiplicité 7, en effet, dans cet exemple le déterminant associé est tres précisément :

$$\frac{36245123848395}{604462909807314587353088}\,\frac{{a_{0,0}}^7\pi^{17}}{y^6}$$

- Voici un exemple où la recherche d'éléments cycliques demande plus de travail. Pour le polynôme $P = X_1\,X_2 \in \mathbf{A}_{p,2}$ et la déformation 3, on obtient :

```
----------
Teste la famille de polynomes
                                    a[0, 0]
Itere 2/4 calcule. Delai:=0.05s.
Itere 3/4 calcule. Delai:=0.134s.
Itere 4/4 calcule. Delai:=0.184s.
Delai de calcul du determinant:=   0.2s.
ECHEC !!
----------
Teste la famille de polynomes
                             a[1, 0] x[1] + a[0, 0]
Itere 2/4 calcule. Delai:=0.133s.
Itere 3/4 calcule. Delai:=0.234s.
Itere 4/4 calcule. Delai:=0.266s.
Delai de calcul du determinant:=   0.233s.
ECHEC !!
----------
Teste la famille de polynomes
                     a[1, 0] x[1] + a[2, 0] x[2] + a[0, 0]
Itere 2/4 calcule. Delai:=0.633s.
Itere 3/4 calcule. Delai:=0.3s.
Itere 4/4 calcule. Delai:=0.35s.
Delai de calcul du determinant:=   0.333s.
ECHEC !!
----------
Teste la famille de polynomes
           a[3, 0] x[1] x[2] + a[1, 0] x[1] + a[2, 0] x[2] + a[0, 0]
Itere 2/4 calcule. Delai:=0.266s.
Itere 3/4 calcule. Delai:=0.317s.
Itere 4/4 calcule. Delai:=0.4s.
Delai de calcul du determinant:=   0.45s.
ECHEC !!
----------
Teste la famille de polynomes
     a[3, 0] x[1] x[2] + a[1, 0] x[1] + a[2, 0] x[2] + a[0, 0] + a[0, 1] y
Itere 2/4 calcule. Delai:=0.666s.
Itere 3/4 calcule. Delai:=0.35s.
Itere 4/4 calcule. Delai:=0.45s.
Delai de calcul du determinant:=   0.483s.
ECHEC !!
----------
Teste la famille de polynomes
   a[3, 0] x[1] x[2] + a[1, 0] x[1] + a[1, 1] y x[1] + a[2, 0] x[2] + a[0, 0]
        + a[0, 1] y
Itere 2/4 calcule. Delai:=0.284s.
Itere 3/4 calcule. Delai:=0.35s.
Itere 4/4 calcule. Delai:=0.45s.
Delai de calcul du determinant:=   0.95s.
```

```
************************************************************
La famille suivante comporte des elements cycliques
 pour la connexion "y Nabla_y".
   a[3, 0] x[1] x[2] + a[1, 0] x[1] + a[1, 1] y x[1] + a[2, 0] x[2] + a[0, 0]
        + a[0, 1] y
Elle est enregistree dans la variable globale de nom "cyclique"
Delai de la recherche:=   8.75s.
************************************************************
                          CONDITIONS DE NON CYCLICITE
{a[0, 0] = a[0, 0], a[1, 0] = a[1, 0], a[0, 1] = a[0, 1], a[3, 0] = a[3, 0],
    a[1, 1] = a[1, 1], a[2, 0] = 0},
{a[0, 0] = a[0, 0], a[1, 0] = a[1, 0], a[0, 1] = a[0, 1], a[3, 0] = a[3, 0],
    a[2, 0] = a[2, 0], a[1, 1] = 0},
{a[1, 0] = a[1, 0], a[0, 1] = a[0, 1], a[3, 0] = a[3, 0], a[1, 1] = a[1, 1],
    a[2, 0] = a[2, 0],
                              6               2  4    2                 6
    a[0, 0] = RootOf((324 y  PPi - 648 PPi  y ) _Z  + (324 a[3, 0] y
                 7                    3                                4
         - 1539 y  PPi a[0, 1] - 4 PPi  a[3, 0] + 1332 a[3, 0] PPi y
                  2          2           2          5         3          3
         + 186 PPi  a[3, 0] y  + 1188 PPi  a[0, 1] y  - 36 PPi  a[0, 1] y ) _Z
                        2  8                      7                      2     2
         + 324 a[0, 1]  y  PPi - 1620 a[0, 1] y  a[3, 0] - 258 a[3, 0]  PPi y
                2        2              2  4          5
         + 4 PPi  a[3, 0]  + 720 a[3, 0]  y  - 3015 PPi y  a[0, 1] a[3, 0]
                 2  3                        3  4        2
         + 60 PPi  y  a[0, 1] a[3, 0] - 36 PPi  y  a[0, 1]
                3                           2  6       2
         - 4 PPi  y a[0, 1] a[3, 0] + 378 PPi  y  a[0, 1] )                  }
************************************************************
```

En particulier, l'élément $\xi = y\,X_1 + X_2 + 1$ est bien cyclique dans $\mathbf{M}(X_1\,X_2, 3)$.

- Enfin, lorsque $P = X_1 + X_2 \in \mathbf{A}_{p,2}$ et en deformation 3, on obtient :

```
----------
Teste la famille de polynomes
                                    a[0, 0]
Itere 2/4 calcule. Delai:=0.05s.
Itere 3/4 calcule. Delai:=0.183s.
Itere 4/4 calcule. Delai:=0.2s.
Delai de calcul du determinant:=   0.217s.
ECHEC !!
----------
Teste la famille de polynomes
                             a[1, 0] x[1] + a[0, 0]
Itere 2/4 calcule. Delai:=0.117s.
Itere 3/4 calcule. Delai:=0.25s.
Itere 4/4 calcule. Delai:=0.717s.
Delai de calcul du determinant:=   0.333s.
************************************************************
La famille suivante comporte des elements cycliques
 pour la connexion "y Nabla_y".
                             a[1, 0] x[1] + a[0, 0]
Elle est enregistree dans la variable globale de nom "cyclique"
Delai de la recherche:=   2.15s.
************************************************************
                          CONDITIONS DE NON CYCLICITE
{a[1, 0] = 0, a[0, 0] = a[0, 0]},
                                             3    2          2
{a[1, 0] = a[1, 0], a[0, 0] = RootOf(12 y _Z  PPi  + 36 y _Z  PPi a[1, 0]
                                         2            2    2             3
                         + (24 y a[1, 0]  + 4 a[1, 0]  PPi ) _Z - a[1, 0]  PPi)
    }
************************************************************
```

L'élément $\xi = X_1$ (de même que $\xi = X_2$, par symétrie) est donc cyclique dans $\mathbf{M}(X_1 + X_2, 3)$.

4.3 Procédure `Maple` pour le calcul des polynômes indiciels et exposants formels

Elle suit de près la démarche théorique rappelée au début de cette section.

```
##########################################################################
###########################    COORDONNEES    ###########################
Poly_indiciel:=proc(Poly) #
```

Procédure destinée au calcul des polynômes indiciels et exposants associés à un élément cyclique `Poly`.

```
local text,temp,tempp,k,j,L_Poly,Der_Poly,A,B,LA,TA,LL,h,
      Den,b,Comp,ldg,tdg,PP,PPinfty,sec,total_time:
global card_base,facteur,Base,liste_var;
 interface(quiet=true);
 total_time:=time():
 etoiles(60,`*`):
 print(`GENERATION DE LA BASE CYCLIQUE`): #
```

Génère la liste `L_Poly` des itérés $(y\nabla_y)^{\mathtt{k}}$(`Poly`), pour k$=0,\ldots,N-1$.

```
 temp:=reduit(Poly):
 L_Poly:=[coordonnees(temp)]:
 for k from 1 to card_base-1 do
  sec:=time():
  temp:=reduit(y*diff(temp,y)+y*facteur*temp):
  L_Poly:=[op(L_Poly),coordonnees(temp)]:
  sec:=temps(time()-sec):
  text:=cat(`Itere `,k+1,`/`,card_base,` calcule. Delai:=`,sec):
  lprint(text):
 od:
 etoiles(60,`*`):
 print(`COEFFICIENTS DES POLYNOMES INDICIELS`): #
```

Calcul des coefficients de la matrice compagnon

```
 lprint(`MATRICE COMPAGNON`):
 temp:=reduit(y*diff(temp,y)+y*facteur*temp):
 Der_Poly:=coordonnees(temp):
 Den:=det(L_Poly):
 Comp:=[]:
 for k from 1 to card_base do
   sec:=time():
   LL:=[]:
   for j from 1 to card_base do
    if (j=k) then
     LL:=[op(LL),Der_Poly]:
    else
     LL:=[op(LL),op(j,L_Poly)]:
    fi:
   od:
   Comp:=[op(Comp),det(LL)/Den]:
   sec:=temps(time()-sec):
   text:=cat(`Coeff `,k,`/`,card_base,
   ` de la matrice compagnon calcule. Delai:=`,sec):
   lprint(text):
 od:
#####DENOMINATUER COMMUN
 sec:=time():
 B:=1:
 for k from 1 to card_base do
  B:=lcm(B,denom(op(k,Comp))):
 od:
########################
 etoiles(60,`-`):
 lprint(`POLYNOMES INDICIELS`): #
```

Calcul des coefficients des polynômes indiciels

```
 A:=map(proc(x,t,z) collect(simplify(x*t),z) end,Comp,B,y):
 A:=[op(A),collect(-B,y)]:
 ldg:=ldegree(op(1,A),y):
 for k from 2 to card_base+1 do
  ldg:=min(ldg,ldegree(collect(op(k,A),y),y)):
 od:
 tdg:=degree(op(1,A),y):
 for k from 2 to card_base+1 do
  tdg:=max(tdg,degree(collect(op(k,A),y),y)):
 od:
 LA:=map(proc(x,d,t) coeff(x,t,d) end,A,ldg,y):
 TA:=map(proc(x,d,t) coeff(x,t,d) end,A,tdg,y):
 PP:=0:
 PPinfty:=0:
 for k from 1 to card_base+1 do
   PP:=PP+op(k,LA)*(Theta)^(k-1):
   PPinfty:=PPinfty+op(k,TA)*(Theta)^(k-1):
 od:
 sec:=temps(time()-sec):
 lprint(`Calcul termine. Delai:=`,sec):
 etoiles(60,`*`): #
```

Affichage des résultats

```
 print(`POLYNOME INDICIEL A L'ORIGINE`):
 print(collect(sort(expand(PP)),Theta)):
```

```
 etoiles(60,`-`):
 if degree(PP,Theta)=0 then
  print(`Pas d'exposants formels a l'origine`)
 else
  text:=`LISTE D'EXPOSANTS FORMELS A L'ORIGINE`:
  print(text):
  print(map(simplify,[solve(PP,Theta)])):
 fi:
 etoiles(60,`*`):
 print(`POLYNOME INDICIEL A L'INFINI`):
 print(collect(sort(expand(PPinfty)),Theta)):
 etoiles(60,`-`):
 if degree(PPinfty,Theta)=0 then
  print(`Pas d'exposants formels a l'infini`)
 else
  text:=`LISTE D'EXPOSANTS FORMELS A L'INFINI`:
  print(text):
  print(map(simplify,[solve(PPinfty,Theta)])):
 fi:
 interface(quiet=false):
 total_time:=temps(time()-total_time):
 lprint(`Delai total de la procedure:=`,total_time):
end:
#########################################################################
macro(exposants=Poly_indiciel(cyclique)): #
#########################################################################
```

Raccourci d'appel de procédure

4.4 Exemples numériques de calcul des polynômes indiciels et exposants formels

- Dans le cas : $P = X^3$ et $m = 8$, on obtient pour le vecteur cyclique $\xi = 1$:

```
************************************************************
                         GENERATION DE LA BASE CYCLIQUE
Itere 2/7 calcule. Delai:=0.05s.
Itere 3/7 calcule. Delai:=0.067s.
Itere 4/7 calcule. Delai:=0.1s.
Itere 5/7 calcule. Delai:=0.15s.
Itere 6/7 calcule. Delai:=0.167s.
Itere 7/7 calcule. Delai:=0.2s.
************************************************************
                      COEFFICIENTS DES POLYNOMES INDICIELS
MATRICE COMPAGNON
Coeff 1/7 de la matrice compagnon calcule. Delai:=0.667s.
Coeff 2/7 de la matrice compagnon calcule. Delai:=0.383s.
Coeff 3/7 de la matrice compagnon calcule. Delai:=0.45s.
Coeff 4/7 de la matrice compagnon calcule. Delai:=0.3s.
Coeff 5/7 de la matrice compagnon calcule. Delai:=0.7s.
Coeff 6/7 de la matrice compagnon calcule. Delai:=0.3s.
Coeff 7/7 de la matrice compagnon calcule. Delai:=0.3s.
------------------------------------------------------------
POLYNOMES INDICIELS
Calcul termine. Delai:=   0.1s.
************************************************************
                         POLYNOME INDICIEL A L'ORIGINE
                              5      2           5
                      6561 PPi  Theta  - 13122 PPi  Theta
------------------------------------------------------------
                     LISTE D'EXPOSANTS FORMELS A L'ORIGINE
                                     [0, 2]
************************************************************
                          POLYNOME INDICIEL A L'INFINI
                  7                  6                  5                   4
  - 16777216 Theta  - 159383552 Theta  - 602406912 Theta  - 1158676480 Theta
                         3                  2
       - 1202622464 Theta  - 653641728 Theta  - 162412288 Theta - 12172160
------------------------------------------------------------
                      LISTE D'EXPOSANTS FORMELS A L'INFINI
                 [-11/4, -1/8, -1/2, -7/8, -5/4, -13/8, -19/8]
Delai total de la procedure:=   5.283s.
```

- Dans le cas : $P = X_1\, X_2$ et $m = 3$, on obtient pour le vecteur cyclique $\xi = y\, X_1 + X_2 + 1$:

```
************************************************************
                         GENERATION DE LA BASE CYCLIQUE
Itere 2/4 calcule. Delai:=0.283s.
Itere 3/4 calcule. Delai:=0.3s.
```

```
Itere 4/4 calcule. Delai:=0.817s.
************************************************************
                      COEFFICIENTS DES POLYNOMES INDICIELS
MATRICE COMPAGNON
Coeff 1/4 de la matrice compagnon calcule. Delai:=0.2s.
Coeff 2/4 de la matrice compagnon calcule. Delai:=0.167s.
Coeff 3/4 de la matrice compagnon calcule. Delai:=0.15s.
Coeff 4/4 de la matrice compagnon calcule. Delai:=0.133s.
------------------------------------------------------------
POLYNOMES INDICIELS
Calcul termine. Delai:=   0.133s.
************************************************************
                         POLYNOME INDICIEL A L'ORIGINE
                                         4
                                 16 PPi  Theta
------------------------------------------------------------
                     LISTE D'EXPOSANTS FORMELS A L'ORIGINE
                                        [0]
************************************************************
                          POLYNOME INDICIEL A L'INFINI
                        4              3              2
           - 19683 Theta  - 59049 Theta  - 56862 Theta  - 17496 Theta
------------------------------------------------------------
                      LISTE D'EXPOSANTS FORMELS A L'INFINI
                              [0, -2/3, -4/3, -1]
Delai total de la procedure:=   3.s.
```

- Enfin, dans le cas : $P = X_1 + X_2$ et $m = 3$, on obtient pour le vecteur cyclique $\xi = X_1 + X_2$:

```
************************************************************
                         GENERATION DE LA BASE CYCLIQUE
Itere 2/4 calcule. Delai:=0.116s.
Itere 3/4 calcule. Delai:=0.25s.
Itere 4/4 calcule. Delai:=0.717s.
************************************************************
                      COEFFICIENTS DES POLYNOMES INDICIELS
MATRICE COMPAGNON
Coeff 1/4 de la matrice compagnon calcule. Delai:=0.134s.
Coeff 2/4 de la matrice compagnon calcule. Delai:=0.067s.
Coeff 3/4 de la matrice compagnon calcule. Delai:=0.083s.
Coeff 4/4 de la matrice compagnon calcule. Delai:=0.1s.
------------------------------------------------------------
POLYNOMES INDICIELS
Calcul termine. Delai:=   0.033s.
************************************************************
                         POLYNOME INDICIEL A L'ORIGINE
                           2      2       2              2
                  - 4 PPi  Theta  - 6 PPi  Theta - 2 PPi
------------------------------------------------------------
                     LISTE D'EXPOSANTS FORMELS A L'ORIGINE
                                   [-1, -1/2]
************************************************************
                          POLYNOME INDICIEL A L'INFINI
                      4           3           2
            - 27 Theta  - 162 Theta  - 357 Theta  - 342 Theta - 120
------------------------------------------------------------
                      LISTE D'EXPOSANTS FORMELS A L'INFINI
                              [-5/3, -4/3, -2, -1]
Delai total de la procedure:=   2.183s.
```

Cet exemple est à souligner par le fait qu'il montre que les exposants formels peuvent être de valuation p-adique négative (pour $p = 2$ dans la situation présente).

Remarquons pour terminer que l'ensemble des exposants à l'origine peut être dans certains cas vide. C'est le cas, par exemple, sur $\mathbf{M}(X_1 - X_2, 3)$.

5. CONCLUSION

Ce rapport succint de notre activité informatique en liaison avec des recherches en mathématique pures, qui a délibérément passé sous silence bien d'aspects théoriques fondamentaux sur lesquels abondent les articles de Mebkhout, s'est également montré avare de précisions quant à la diversité des programmes de calcul formel que nous avons élaborés. Nous nous sommes, en effet, limités a donner un aperçu rapide des motivations qui nous ont emmenées à envisager sérieusement d'utiliser l'informatique comme outil de découverte. Conscients du fait que des structures mathématiques riches et profondes non encore dévoilées mais soupçonnées par le mathématicien, demandent de plus en plus souvent des calculs inaccessibles à l'être humain pour être mises en évidence, et sachant que l'informatique, utilisée à bon escient, élargit incontestablement, aujourd'hui, l'horizon du chercheur, nous avons consciemment voulu tester la puissance de cet outil de recherche. Notre expérience des quatre dernières années s'est avérée, en ce sens, extrêmement positive et des percées théoriques fondamentales ont pu être réalisées dans le domaine de la théorie des équations différentielles p-adiques.

Nous persistons à croire en l'importance potentielle pour la recherche scientifique, des Centres de "référence" de Calcul de niveau international, spécifiques pour les Mathématiques, ouverts aux demandes de tous les mathématiciens et leur offrant des installations informatiques de pointe.

Le Centre de Calcul du **GDR-Médicis** en région parisienne, répond dans une large mesure, à l'heure actuelle et de notre point de vue, à cette exigence. Nous souhaitons que Médicis continue d'évoluer dans cette direction tout en se dotant des derniers progrès technologiques.

Alberto Arabia
mercredi 15 mars 2000